

```

1 ;*****
2 ;      RSX : "DISCTAPE"      *
3 ;      (c) by Guido Trier    *
4 ;      1986/Bonn             *
5 ;      Assembler-Listing     *
6 ;*****
7 ;
8 ;
9 ;
10      org    #a100                ;Programmanfang auf #a100
20 buffer: equ    #4000            ;Buffer auf #4000
30      ld     bc,comand           ;Adresse des Kommando in BC
40      ld     hl,kern            ;Adresse des Kern in HL
50      call   #bcd1              ;Einbinden des RSX-Befehl
60      ret
70 kern:  defs   #0006             ;Vom System benutzter Kern fuer RSX
80 comand: defw  tabell           ;Adresse der Kommandotabelle
90      jp     routin            ;Sprungtabelle
100 tabell: defm "DISCTAP"        ;Name des RSX
110      defb   #c5               ;letzter Buchstabe+#80
120      defb   0                 ;Kennzeichen fuer Ende der Tabelle
125 ;===== PARAMETER-UEBERGABE =====
130 routin: ld     h,(ix+#01)
140      ld     l,(ix+#00)
160      ld     (#ac00),hl        ;Geschwindigkeit nach #ac00
170      ld     h,(ix+#03)
180      ld     l,(ix+#02)
181      ld     a,(hl)
182      ld     (#ac25),a         ;Laenge des Kassettennamen nach #ac25
185      inc    hl
190      ld     e,(hl)
200      inc    hl
210      ld     d,(hl)
220      ld     (#ac20),de        ;Adresse des Kassettennamen nach #ac20
230      ld     h,(ix+#05)
240      ld     l,(ix+#04)
241      ld     a,(hl)
242      ld     (#ac15),a         ;Laenge des Diskettennamen nach #ac15
245      inc    hl
250      ld     e,(hl)
260      inc    hl
270      ld     d,(hl)
280      ld     (#ac10),de        ;Adresse des Diskettennamen nach #ac10
285 ;===== UMSCHALTEN AUF DISC =====
290      ld     hl,disc
300      call   #bcd4             ;suchen nach Befehl DISC
310      xor     a                ;kein Parameter ist uebergeben
320      call   #001b            ;ausfuehren des Befehls DISC
325 ;===== HEADER LESEN UND PROGRAMM LADEN =====
330      ld     hl,(#ac10)        ;Adresse des Diskettennamen in HL
331      ld     a,(#ac15)
340      ld     b,a              ;Laenge des Diskettenname in B
350      ld     de,buffer        ;Adresse des Buffers in DE
360      call   #bc77            ;DISC IN OPEN
370      ld     (#ac30),de        ;Anfangsadresse des Programm nach #ac30
380      ld     (#ac40),bc        ;Laenge des Programms nach #ac40
390      ld     (#ac60),a         ;Art des Programms nach #ac60
400      ld     hl,(#ac30)        ;Anfangsadresse des Programm in HL
410      call   #bc83            ;Programm laden
420      ld     (#ac50),hl        ;Einsprungsadresse nach #ac50
430      call   #bc7a            ;DISC IN CLOSE
435 ;===== UMSCHALTEN AUF TAPE =====
440      ld     hl,tape
450      call   #bcd4             ;suchen nach Befehl TAPE
460      xor     a                ;kein Parameter ist uebergeben
470      call   #001b            ;ausfuehren des Befehls TAPE
475 ;===== GESCHWINDIGKEIT SETZEN =====
480      ld     hl,(#ac00)        ;Geschwindigkeit in HL
490      ld     a,50              ;Vorprueflaenge in A
500      call   #bc68            ;Geschwindigkeit setzen
501 ;===== PROGRAMM SPEICHERN =====
505      ld     a,(#ac25)
510      ld     b,a              ;Laenge des Kassettename in B
520      ld     hl,(#ac20)        ;Adresse des Kassettename in HL
530      ld     de,buffer        ;Adresse des Buffers in DE
540      call   #bc8c            ;CAS OUT OPEN
550      ld     hl,(#ac30)        ;Anfangsadresse des Programms in HL
560      ld     de,(#ac40)        ;Laenge des Programms in DE
570      ld     bc,(#ac50)        ;Einsprungsadresse des Programms in BC
580      ld     a,(#ac60)        ;Programmart in a
590      call   #bc98            ;Programm speichern
600      call   #bc8f            ;CAS OUT CLOSE
605 ;===== UMSCHALTEN AUF DISC =====
610      ld     hl,disc
620      call   #bcd4             ;suchen nach Befehl DISC
630      xor     a                ;kein Parameter ist uebergeben
640      call   #001b            ;ausfuehren des Befehls DISC
650      ret
660 disc:  defm "DIS"             ;Name fuer DISC
670      defb   #c3               ;letzter Buchstabe+#80
680 tape:  defm "TAP"            ;Name fuer TAPE
690      defb   #c5               ;letzter Buchstabe+#80

```